/*
 * Copyright © 2019 Apple Inc. and the ServiceTalk project authors
 *
 * Licensed under the Apache License, Version 2.0 (the "License");
 * you may not use this file except in compliance with the License.
 * You may obtain a copy of the License at
 *
 *   http://www.apache.org/licenses/LICENSE-2.0
 *
 * Unless required by applicable law or agreed to in writing, software
 * distributed under the License is distributed on an "AS IS" BASIS,
 * WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
 * See the License for the specific language governing permissions and
 * limitations under the License.
 */

apply plugin: "io.servicetalk.servicetalk-gradle-plugin-internal-library"

// Required version for module
def javaLanguageVersion = JavaVersion.VERSION_1_9

if (!JavaVersion.current().isCompatibleWith(javaLanguageVersion)) {
    project.tasks.all { task -> task.enabled = false }
}

java {
    sourceCompatibility = javaLanguageVersion
    targetCompatibility = javaLanguageVersion
}

compileJava {
    options.release = Integer.parseInt(javaLanguageVersion.getMajorVersion())
}

compileTestJava {
    options.release = compileJava.options.release
}

dependencies {
    api project(":servicetalk-concurrent")
    api project(":servicetalk-concurrent-api")

    implementation project(":servicetalk-annotations")
    implementation project(":servicetalk-concurrent-api-internal")

    testImplementation enforcedPlatform("org.junit:junit-bom:$junit5Version")
    testImplementation testFixtures(project(":servicetalk-concurrent-api"))
    testImplementation testFixtures(project(":servicetalk-concurrent-internal"))
    testImplementation project(":servicetalk-test-resources")
    testImplementation "org.junit.jupiter:junit-jupiter-api"
    testImplementation "org.mockito:mockito-core:$mockitoCoreVersion"
    testImplementation "org.hamcrest:hamcrest:$hamcrestVersion"
}
